

Volume 20

April , 1990

Bulletin 1

Page 1

A SIMPLE COMPUTER PROGRAM FOR OBTAINING
WEIBULL PARAMETERS FROM FAILURE DATA
WITHOUT ANY SORTING INTO ORDER STATISTICS

INTRODUCTION

Ever since the Weibull distribution was first used extensively (in the early 1950's) to analyze data on fatigue specimens or other life tests it has been a common practice to use order statistics and regression analysis on Weibull plots. The rearranging of measured values into numerical order can be a big nuisance in case we have large samples and we want to enter a data set just as it comes, especially if we desire quick estimates of Weibull slope and characteristics life in small personal or pocket computers with which we can use the BASIC programming language. The purpose of this bulletin is to introduce just such a simple program for practical use in estimating Weibull parameters. It is an empirical approach which has stood the test of time over many years of experimentation. After seeing how amazingly close these empirical Weibull parameters are to the truly rigorous values in any problems, the user will be fully convinced about the practical value of this simplified procedure.

THE TWO EMPIRICAL RULES WHICH FORM THE BASIS OF THE PROGRAM

The basis of our simple Weibull program consists of two empirical rules. These are

EMPIRICAL RULE # 1: The Weibull slope is inversely proportional to the standard deviation of the natural logarithms of the observed measurements, i.e.,

$$b = K / \text{Sigma}_{\ln X} ,$$

where X = Measured Value's Symbol

$\ln X$ = Natural Logarithms of Measured Value

$\text{Sigma}_{\ln X}$ = Standard Deviation of the Natural
Logarithms of the Data Values
(i.e., Measured Values)

b = Weibull Slope

K = A Constant Depending on the Sample Size

If N = Sample Size, then the empirical formula for K is

$$K = 1.4 \{1 - \exp[-(N/4)^{.3125}]\}$$

EMPIRICAL RULE # 2: The natural logarithm of the characteristic life in the Weibull distribution is the MEAN VALUE of the LOGARITHMS of the measured data values (lives) PLUS 52% of the Standard Deviation of the Logarithms of the measured values. Expressed mathematically this says that

$$\ln \text{theta} = M_{\ln X} + .52(\text{Sigma}_{\ln X})$$

where **theta** = Characteristic Life

$M_{\ln X}$ = Mean of the Natural Logarithms
of the data values

$\text{Sigma}_{\ln X}$ = Standard Deviation of the
Natural Logarithms of the data
values

So, as a result, we have the following empirical formula for the Characteristic Life :

$$\text{theta} = \exp[M_{\ln X} + .52(\text{Sigma}_{\ln X})]$$

NOTE: Because of the use of logarithms of the data values we name this program "LOG-WEIBULL".

THE basic PROGRAM GENERATED FROM THE TWO EMPIRICAL RULES

```
10 PRINT TAB(30)"LOG WEIBULL PROGRAM"
15 PRINT
20 PRINT
30 N = 0
40 T = 0
50 L = 99999999#
60 D = 0
70 H = -1
75 PRINT TAB(20)"GIVE DATA VALUES"
80 INPUT Q
85 IF Q > 999999! THEN 110
90 PRINT TAB(20)"VALUE=";Q
102 IF Q<L THEN L=Q
104 IF Q>H THEN H=Q
110 Q = LOG(Q)
120 IF Q>13 THEN 190
150 N = N+1
160 T = T+Q
170 D = D+Q*Q
180 GOTO 80
190 M = T/N
200 V = (D/N) - M*M
210 S = SQR(V)
220 W = (N/4)^.3125
230 K = 1.4*(1 - EXP(-W))
240 B = K/S
250 X = M + .52*S
260 Y = EXP(X)
270 PRINT
275 PRINT
280 PRINT TAB(20)"SAMPLE SIZE=";N
290 PRINT TAB(20)"RANGE=";H-L
310 PRINT TAB(20)"LOWEST=";L
320 PRINT TAB(20)"HIGHEST=";H
350 PRINT TAB(20)"WEIBULL SLOPE=";B
370 PRINT TAB(20)"CHARACTERISTIC LIFE=";Y
390 END
```

A NUMERICAL EXAMPLE

Five ball bearings were tested to failure with the following results:

Life of 1st bearing tested = 150 Hrs.

Life of 2nd bearing tested = 97 Hrs.

Life of 3rd bearing tested = 220 Hrs.

Life of 4th bearing tested = 300 Hrs.

Life of 5th bearing tested = 51 Hrs.

We enter the data values one at a time into the computer, and obtain the computer printout shown on the next page. NOTE: After entering the last data item (51 Hrs. in this case), we enter the **LARGE FICTITIOUS VALUES** of 1000000 (one million) in order to signal the end of the data set. Then the program completes its summations and Weibull parameter estimates

REMEMBER: This program cannot be used if there are suspended (unfailed) items.

LOG WEIBULL PROGRAM

GIVE DATA VALUES

VALUE= 150

VALUE= 97

VALUE= 220

VALUE= 300

VALUE= 51

SAMPLE SIZE= 5

RANGE= 249

LOWEST= 51

HIGHEST= 300

WEIBULL SLOPE= 1.477382

CHARACTERISTIC LIFE= 190.0017

CONCLUSIONS

1. In the example the MEDIAN RANK PLOT on WEIBULL PAPER verifies the parameters in the computer printout.
2. This computer program is very handy for any technician or engineer who is in a hurry to get needed Weibull Parameter estimates for tests in which each specimen is run to failure.